

collect stars — Add stars for significant results in a collection

Description
Options

Quick start
Remarks and examples

Menu
Stored results

Syntax
Also see

Description

`collect stars` manages the creation of stars for indicating the significance of results in a collection.

Quick start

Specify three levels of stars based on p -values in `_r_p` in the `result` dimension

```
collect stars _r_p 0.01 "***" 0.05 "***" 0.1 "*"
```

As above, and attach the stars to coefficients stored in `_r_b`

```
collect stars _r_p 0.01 "***" 0.05 "***" 0.1 "*", attach(_r_b)
```

Menu

Statistics > Summaries, tables, and tests > Tables and collections > Build and style table

Syntax

```
collect stars resultID
  [#1 "label1"
  [#2 "label2"
  [#3 "label3"
  [#4 "label4"
  [#5 "label5" ]]]]] [ , options ]
```

resultIDs are levels in the `result` dimension whose values determine the stars to be applied.

Value-label pairs are rearranged such that $\#1 \leq \#2 \leq \#3 \leq \#4 \leq \#5$.

For value v corresponding to one of the results in *resultIDs*,

- if $v > \#5$, no new stars value is created
- if $v \leq \#5$, the new stars value is set to "*label5*"
- if $v \leq \#4$, the new stars value is set to "*label4*"
- if $v \leq \#3$, the new stars value is set to "*label3*"
- if $v \leq \#2$, the new stars value is set to "*label2*"
- if $v \leq \#1$, the new stars value is set to "*label1*"

2 collect stars — Add stars for significant results in a collection

<i>options</i>	Description
Main	
<code>name(<i>cname</i>)</code>	specify stars for collection <i>cname</i>
<code>attach(<i>attachres</i>)</code>	specify that stars be appended to items in <i>attachres</i>
Options	
<code>fortags(<i>taglist</i>)</code>	specify tags identifying items for which to generate new star items

Options

Main

`name(cname)` specifies the collection to which stars are to be applied. By default, stars are applied to the current collection.

`attach(attachres)` specifies that items with result levels in *attachres* be appended with the associated star label when rendered to a table.

Options

`fortags(taglist)` specifies conditions for selecting which values are to be used to create stars. Values with tags in *taglist* will be used to generate new stars.

Within the *taglist*, if tags are joined by #, values having all of these tags are selected; if tags are separated by a space, values with any of these tags are selected.

taglist contains

tagspec

tagspec taglist

tagspec contains

tag

tag#tag[#tag[...]]

tag contains

dimension

dimension[*levels*]

dimension is a dimension in the collection.

levels are levels of the corresponding dimension.

Distinguish between `[]`, which are to be typed, and `[]`, which indicate optional arguments.

Remarks and examples

[stata.com](https://www.stata.com)

Stars are often used in tables to denote significance. `collect stars` allows you to include stars in your table based on other values, typically *p*-values that are already in your collection.

To demonstrate, we first create a table of regression results displaying coefficients and *p*-values.

```

. use https://www.stata-press.com/data/r17/nhanes2
. quietly: collect: regress bpsystol bmi i.region age
. collect style showbase off
. collect layout (colname) (result[_r_b _r_p])
Collection: default
  Rows: colname
  Columns: result[_r_b _r_p]
Table 1: 6 x 2

```

	Coefficient	p-value
Body mass index (BMI)	1.303719	0.000
MW	-.0659707	0.907
S	-.5170085	0.356
W	-.6045511	0.289
Age (years)	.5887217	0.000
Intercept	69.89029	0.000

Rather than showing the p -values, we can use `collect stars` to define the levels of the p -values stored in `_r_p` for which stars should be shown. Here we will use three stars for values less than 0.01, two stars for values less than 0.05, and one star for values less than 0.1. A new `stars` level in the `result` dimension is created and can be used in our table layout.

```

. collect stars _r_p 0.01 "****" 0.05 "***" 0.1 "*"
. collect layout (colname) (result[_r_b stars])
Collection: default
  Rows: colname
  Columns: result[_r_b stars]
Table 1: 6 x 2

```

	Coefficient	stars
Body mass index (BMI)	1.303719	***
MW	-.0659707	
S	-.5170085	
W	-.6045511	
Age (years)	.5887217	***
Intercept	69.89029	***

It is unlikely that we want the level name `stars` to show in the column header. It would also be helpful to left align the stars to be closer to the reported coefficients. We can do this with `collect style header` and `collect style cell`.

```

. collect style header result[stars], level(hide)
. collect style cell result[stars], halign(left)
. collect preview

```

	Coefficient
Body mass index (BMI)	1.303719 ***
MW	-.0659707
S	-.5170085
W	-.6045511
Age (years)	.5887217 ***
Intercept	69.89029 ***

4 collect stars — Add stars for significant results in a collection

Alternatively, we can directly attach the stars to the coefficient by specifying the `attach()` option and naming the result (`_r_b`) that we want the stars attached to.

```
. collect stars _r_p 0.01 "****" 0.05 "** " 0.1 "* " 1 " ", attach(_r_b)
. collect layout (colname) (result[_r_b])
Collection: default
  Rows: colname
  Columns: result[_r_b]
Table 1: 6 x 1
```

	Coefficient
Body mass index (BMI)	1.303719***
MW	-.0659707
S	-.5170085
W	-.6045511
Age (years)	.5887217***
Intercept	69.89029***

Here we added extra spaces to force three characters following the number in each cell. This gives nice alignment when we are looking at the results in formats such as plain text and the Stata Markup and Control Language format. However, if you are exporting your table to other formats, you may prefer to leave the stars in a separate column and apply alignment and margin styles to achieve your desired look.

Stored results

`collect stars` stores the following in `s()`:

```
Macros
  s(collection)  name of collection
  s(value1)      star cutoff value 1
  s(label1)      star label 1
  s(value2)      star cutoff value 2
  s(label2)      star label 2
  s(value3)      star cutoff value 3
  s(label3)      star label 3
  s(value4)      star cutoff value 4
  s(label4)      star label 4
  s(value5)      star cutoff value 5
  s(label5)      star label 5
```

Also see

[TABLES] [collect clear](#) — Clear all collections in memory

[TABLES] [collect drop](#) — Drop collections from memory